Limita posloupnosti

Limita posloupnosti

Tomáš Kasalický, 2014

Připomenutí pojmů

Demonstrujme pojem limity posloupnosti, nejprve ale připomeňme samotnou definici:

Řekneme, že reálná posloupnost (a_n) má limitu $\alpha \in \mathbb{R}$, právě když pro každé okolí H_α bodu α lze nalézt $n_0 \in \mathbb{N}$ větší než n_0 tak, že platí $a_n \in H_\alpha$. V symbolech

$$(\forall H_\alpha)(\exists n_0 \in \mathbb{N})(\forall n \in \mathbb{N})(n > n_0 \Rightarrow a_n \in H_\alpha).$$

Připomeňme také definici okolí bodů z rozšířené reálné osy $\bar{\mathbb{R}}$. Pokud $a \in \mathbb{R}$, pak velikost okolí je udána kladným reálným parametrem ϵ,

$$H_a(\epsilon) = (a - \epsilon, a + \epsilon).$$

Pokud $a = +\infty$, pak pro $c \in \mathbb{R}$ klademe

$$H_{+\infty}(c) = (c, +\infty).$$

Podobně v případě $a = -\infty$ máme pro $c \in \mathbb{R}$

$$H_{-\infty}(c) = (-\infty, c).$$

$\alpha \in \mathbb{R}$

Názorně demonstrujme tvrzení $\lim\limits_{n \to \infty} \dfrac{1}{n} = 0$. Tento příklad je ukázkou konvergentní posloupnosti, tedy posloupnosti mající konečnou limitu $(0 \in \mathbb{R})$.

```
def seq_plot(epsilon,n0):
    a(n)=1/n
    graph = Graphics()
    graph += list_plot([a(i) for i in [1..10]])
    graph += polygon([(0,0 - epsilon), (10,0 - epsilon),
```

```
(10,0 + epsilon),(0,0 + epsilon)], color = "red", alpha = .3)
    graph += polygon([(n0,0 - 1), (10,0 - 1), (10,0 +
1),(n0,0 + 1)], color = "green", alpha = .3)
    show (graph)
#####################
# make Interaction
######################
@interact
def _( epsilon= slider(srange(.05,1.001,.01),default = .5,
    label="$\epsilon$ :    "),n0=
slider(srange(1,11,1),default = 2,
    label="$n_0$ :    ") ):
    seq_plot(epsilon,n0)
```

ϵ : 0.500000000000000

n_0 : 2



Tuto limitu si můžeme snadno nechat ověřit přímo pomocí Sage:

```
show(limit(1/n,n=infinity))
```

0

Další ukázkou konvergentní posloupnosti může být posloupnost daná předpisem $a_n = \frac{5n}{n+1}, n = 1, 2, 3, \ldots$

```
def seq_plot(epsilon,n0):
    a(n)= 5*n/(n+1)
    alpha = 5

    graph = Graphics()
    graph += list_plot([a(i) for i in [1..40]])
```

```
    graph += polygon([(0,alpha - epsilon), (40,alpha -
epsilon), (40,alpha + epsilon),(0,alpha + epsilon)], color =
"red", alpha = .3)
    graph += polygon([(n0,0), (40,0), (40,alpha +
1),(n0,alpha + 1)], color = "green", alpha = .3)
    graph += line([(0,alpha),
(40,alpha)],linestyle='dashed',color='gray')
    show (graph,aspect_ratio=3)
#####################
# make Interaction
######################
@interact
def _( epsilon= slider(srange(.15,1.001,.01),default = .5,
    label="$\epsilon$ :    "),n0=
slider(srange(1,41,1),default = 10,
    label="$n_0$ :    ") ):
    seq_plot(epsilon,n0)
```

ϵ : 0.500000000000000

n_0 : 10



Dle Sage je její limitou skutečně 5.

```
show(limit(5*n/(n+1), n=+Infinity))
```

5

Podívejme se ještě na jednu posloupnost s konečnou limitou. Ukážeme, že $\lim_{n\to\infty} \frac{\sin(n)}{n} = 0$.

Tato posloupnost je ze začátku podobná sinusoidě, která nemá limitu, ale postupně se přibližuje ose n a má tak limitu rovnu 0.

```
def seq_plot(epsilon,n0):
    a(n)=sin(n)/n
    graph = Graphics()
    graph += list_plot([a(i) for i in [1..30]])
    graph += polygon([(0,0 - epsilon), (30,0 - epsilon),
(30,0 + epsilon),(0,0 + epsilon)], color = "red", alpha = .3)
    graph += polygon([(n0,0 - 1), (30,0 - 1), (30,0 +
1),(n0,0 + 1)], color = "green", alpha = .3)
    show (graph,aspect_ratio=10)
#####################
# make Interaction
######################
@interact
def _( epsilon= slider(srange(.05,1.001,.01),default = .5,
    label="$\epsilon$ :    "),n0=
slider(srange(1,31,1),default = 10,
    label="$n_0$ :    ") ):
    seq_plot(epsilon,n0)
```

ϵ : 0.500000000000000

n_0 : 10




```
show(limit(sin(n)/n, n=+infinity))
```

0

$\alpha = +\infty$

Demonstrujme tento jev na posloupnosti (n^2). Tvrdíme tedy, že $n^2 \to +\infty$.

```
def seq_plot(c,n0):
    a(n)=n^2
    graph = Graphics()
    graph += list_plot([a(i) for i in [1..50]])
```

```
    graph += polygon([(0,c), (50,c), (50,2500),(0,2500)],
color = "red", alpha = .3)
    graph += polygon([(n0,0), (50,0), (50,2500),(n0,2500)],
color = "green", alpha = .3)
    show (graph,aspect_ratio=.02)
#####################
# make Interaction
######################
@interact
def _( c= slider(srange(1,2501,1),default = 1000,
    label="$c$ :    "),n0= slider(srange(1,51,1),default = 20,
    label="$n_0$ :    ") ):
    seq_plot(c,n0)
```

c : 1002

n_0 : 20



Podobně jako v předchozím případě, Sage požadovanou limitu snadno spočte.

```
show(limit(n^2, n=+infinity))
```

$+\infty$

Tato posloupnost nemá konečnou limitu. Posloupnosti s touto vlastností nazýváme divergentní.

Posloupnost nemající limitu

Tento příklad ukazuje, že existují posloupnosti nemající limitu (ani konečnou, ani nekonečnou). To by nemělo být překvapivé. Uvažme posloupnost zadanou předpisem $a_n = \sin(n)$, $n = 1, 2, 3, \ldots$

```
def seq_plot(epsilon,n0):
    a(n)=sin(n)
    graph = Graphics()
    graph += list_plot([a(i) for i in [1..30]])
    graph += polygon([(0,0 - epsilon), (30,0 - epsilon),
(30,0 + epsilon),(0,0 + epsilon)], color = "red", alpha = .3)
    graph += polygon([(n0,0 - 1), (30,0 - 1), (30,0 +
1),(n0,0 + 1)], color = "green", alpha = .3)
    show (graph,)
#####################
# make Interaction
######################
@interact
def _( epsilon= slider(srange(.05,1.001,.01),default = .5,
    label="$\epsilon$ :    "),n0=
slider(srange(1,31,1),default = 10,
    label="$n_0$ :    ") ):
    seq_plot(epsilon,n0)
```

ϵ : 0.500000000000000

n_0 : 10



Po několika experimentech s nastavením parametrů v předchozím obrázku vidíme, že pro některá $\epsilon > 0$ nelze nalézt v definici pořadovná n_0. Tato posloupnost tedy limitu nemá. Co na to říká Sage?

```
show(limit(sin(n), n=+Infinity))
```

ind

"ind" znamená "indeterminate". Sage si tedy neví rady, limita pravděpodobně neexistuje (CAS nelze věřit).

Poznámka k výpočtu limit pomocí Sage

Při výpočtu limit pomocí Sage (ale i Mathematica!) bývá nutné explicitně udat, že počítáme limitu posloupnosti. Sice jsme si ve výpočtech výše označili proměnnou jako n, ale Sage ve skutečnosti počítá limitu funkce závislé na spojité proměnné n. Pokud tato limita existuje, pak je stejná s limitou posloupnosti (uvidíme běheme semestru, viz Heineho věta). To ovšem může vést k problémům. Uvažme posloupnost

$$a_n = \sin(\pi n), \quad n = 1, 2, 3, \ldots$$

Tato posloupnost je ve skutečnosti konstantní $a_n = 0$ a má tedy limitu 0. Sage nám ale vrací:

```
show(limit(sin(pi*n), n=+infinity))
```

$$ind$$

Což je nás překvapí, protože jak bylo řečeno, limita existuje! Důvodem je, že Sage se na zadaný výraz dívá jako na limitu funkce $\sin(\pi x)$ a ta, jakožto nekonstatntní periodická funkce, v nekonečnu samozřejmě limitu nemá. Musíme tedy Sage donutit předpokládat, že n je celočíselná proměnná. To lze snadno.

```
var('n')
assume(n,'integer')
```

```
show(limit(sin(pi*n), n=+infinity))
```

$$0$$

Tento výsledek nás již uspokojuje.